Title stata.com

collect label — Manage custom labels in a collection

Description Quick start Menu

Syntax Options Remarks and examples

Stored results References Also see

# **Description**

collect label dim defines the label for a dimension. If the label is specified as an empty string, any existing label for the dimension is removed.

collect label levels defines labels for the levels of a dimension. If the label is specified as an empty string, any existing label for the level is removed.

collect label save saves label definitions to the specified file.

collect label use adds the labels defined in an external file to a collection.

collect label drop drops the dimension label and the labels for its levels.

collect label list lists the dimension label and the labels for its levels.

All collect label subcommands operate on the current collection by default.

### **Quick start**

In the current collection, label dimension dim1 as "First dimension", overwriting the default label collect label dim dim1 "First dimension", modify

Clear the label for dimension dim1

collect label dim dim1 ""

Redefine the labels for the levels of dimension dim1

collect label levels dim1 level1 "Level 1" level2 "Level 2", modify

Save all defined labels from the current collection to mylabels.stjson for use with other collections collect label save mylabels

Same as above, but save labels from the collection c2 to mylabels2.stjson collect label save mylabels2, name(c2)

Drop the dimension label and level labels from dimension dim1 collect label drop dim1

List all labels for dimension dim1

collect label list dim1

### Menu

#### collect label dim

Statistics > Summaries, tables, and tests > Tables and collections > Build and style table

#### collect label levels

Statistics > Summaries, tables, and tests > Tables and collections > Build and style table

#### collect label save

Statistics > Summaries, tables, and tests > Tables and collections > Collection labels > Save labels

#### collect label use

Statistics > Summaries, tables, and tests > Tables and collections > Collection labels > Use labels

### collect label drop

Statistics > Summaries, tables, and tests > Tables and collections > Collection labels > Drop labels

#### collect label list

Statistics > Summaries, tables, and tests > Tables and collections > Collection labels > List labels

## **Syntax**

```
Label a dimension
```

```
collect label dim dim "label" [ , name(cname) modify]
```

#### Label dimension levels

```
collect label levels dim level "label" [level "label" ...]
[, name(cname) modify replace]
```

Save labels to disk

```
collect label save filename [ , name(cname) replace ]
```

Use labels from disk

```
collect label use filename [ , name(cname) modify replace ]
```

Drop labels for a dimension

```
collect label drop dim [ , name(cname) ]
```

List labels for a dimension

```
collect label list dim [ , name(cname) all]
```

where *cname* is a collection name, *dim* is a dimension in the specified collection, and *level* is a level of this dimension.

If filename is specified without an extension, .stjson is assumed for both collect label save and collect label use. If filename contains embedded spaces, enclose it in double quotes.

# **Options**

Options are presented under the following headings:

Options for collect label dim Options for collect label levels Options for collect label save Options for collect label use Option for collect label drop Options for collect label list

### Options for collect label dim

name (cname) specifies the collection to which the dimension label is to be applied. By default, the dimension label is applied to the current collection.

modify specifies that an existing (nonempty) dimension label is to be modified. By default, collect label dim applies labels only to dimensions with empty labels.

### Options for collect label levels

name (cname) specifies the collection to which the level labels are to be applied. By default, the level labels are applied to the current collection.

modify specifies that existing (nonempty) labels on dimension levels are to be modified. By default, new labels specified using collect label levels are applied only to levels without existing labels, and those with existing labels will remain unchanged.

replace specifies that all existing labels for the levels of dimension dim be dropped and only the new level labels be applied.

## Options for collect label save

name (cname) specifies the collection from which labels are to be saved. By default, the labels from the current collection are saved.

replace permits collect label save to overwrite an existing file.

## Options for collect label use

name (cname) specifies the collection to which the labels in *filename* will be applied. By default, the labels in *filename* will be applied to the current collection.

modify specifies that existing (nonempty) labels on dimensions and dimension levels are to be modified. By default, the labels in *filename* are applied only to dimensions and levels of dimensions without existing labels, and those with existing labels remain unchanged.

replace specifies that all existing labels on dimensions and dimension levels be dropped and only the new labels in *filename* be used.

### Option for collect label drop

name(cname) specifies the collection for which labels are to be dropped. By default, labels are dropped for the dimension in the current collection.

### Options for collect label list

name(cname) specifies the collection for which labels are to be listed. By default, labels are listed for the dimension in the current collection.

all specifies that all levels of dim be shown in the list, including those without a label.

# Remarks and examples

stata.com

Labels of dimensions and their levels allow us to quickly understand what values are being presented in a table.

When you use collect get or the collect prefix, some labels are automatically applied to dimensions and levels of dimensions in your collection. When you have variable and value labels on the variables in your dataset, those labels will be included in your collection as well, and they will be displayed in tables created from it. In addition, Stata provides default labels for dimensions such as the result dimension, which means that the statistics reported on the table will have meaningful labels. However, for a given table, it is often necessary to modify labels. For instance, value labels from the dataset may be longer than desired for a column header, or you may prefer a different description than the default provided by Stata for the statistics in your table. collect label allows you to make these changes.

To demonstrate, we use data from the Second National Health and Nutrition Examination Survey (NHANES II) (McDowell et al. 1981) to fit two models. We collect the coefficients (\_r\_b) and use the quietly prefix to suppress the output. Then, we arrange the items in our collection with collect layout. We place the variable names on the rows and the statistics (result) from each command (cmdset) on the columns:

. use https://www.stata-press.com/data/r18/nhanes2

. quietly: collect \_r\_b: regress bpsystol bmi

. quietly: collect  $\_r\_b$ : regress bpsystol bmi age

. collect layout (colname) (cmdset#result)

Collection: default Rows: colname Columns: cmdset#result

Table 1:  $3 \times 2$ 

	1 Coefficient	2 Coefficient
Body mass index (BMI) Age (years) Intercept	1.656894	1.304128 .5883367 69.58451

Each covariate is a level of the dimension colname, and the levels of colname are labeled with the variable labels from the current dataset in memory, by default. The constant is labeled as Intercept. Say that we want to change that label to Constant. We can use collect label levels to change the label. We will first need to know the name of this level within the colname dimension. To determine the name, we can simply list all the levels of the dimension and their corresponding labels as follows:

```
. collect label list colname

Collection: default
Dimension: colname
    Label: Covariate names and column names
Level labels:
    _cons    Intercept
        age    Age  (years)
        bmi    Body mass index  (BMI)
```

Here we see the label for the dimension as well as each of its levels. By default, the dimension label is omitted from the table. We find that the name of the level for the constant is \_cons. Because this level already has a label, we need the modify option with collect label levels to override the existing label. In the table above, we also see 1 and 2 in the column headers. These identify our two regression commands; they are the levels of the cmdset dimension, which are unlabeled. We could learn this by typing

. collect label list cmdset, all

We will label the columns as Model 1 and Model 2. Because the levels of cmdset do not have labels, we do not need to specify any options. After applying our labels, we get a preview of the table:

```
. collect label levels colname _cons "Constant", modify . collect label levels cmdset 1 "Model 1" 2 "Model 2"
```

. collect preview

	Model 1 Coefficient	Model 2 Coefficient
Body mass index (BMI) Age (years)	1.656894	1.304128 .5883367
Constant	88.56855	69.58451

There may be certain labels that you find yourself modifying for many tables. For example, if you create many tables of estimation results, you might change the label Coefficient to simply Coef. in each table. Instead of doing this for each collection separately, you can save a set of labels with collect label save. Then, you can use collect label use with option modify to apply these labels to other collections.

## Stored results

All collect label subcommands store the following in s():

```
Macros
s(collection) name of collection
s(dimname) dimension name
```

collect label list additionally stores the following in s():

```
Macros
s(label) dimension label
s(level#) level of dimension
s(label#) label for level of dimension
s(k) number of dimension levels with a label
```

### References

Huber, C. 2021. Customizable tables in Stata 17, part 2: The new collect command. The Stata Blog: Not Elsewhere Classified. https://blog.stata.com/2021/06/07/customizable-tables-in-stata-17-part-2-the-new-collect-command/.

McDowell, A., A. Engel, J. T. Massey, and K. Maurer. 1981. Plan and operation of the Second National Health and Nutrition Examination Survey, 1976-1980. Vital and Health Statistics 1(15): 1-144.

#### Also see

[TABLES] collect dims — List dimensions in a collection [TABLES] collect levels of a dimension

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985-2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on citing Stata documentation.